dnl
dnl LC_CHECK_TYPE_SIZE(TYPENAME [, INCLUDES=INCLUDES-DEFAULT] [, IF-ERROR])
dnl
AC_DEFUN([LC_CHECK_TYPE_SIZE],
  [AC_PREREQ([2.59])dnl
   AS_VAR_PUSHDEF([CACHEVAR], [lc_cv_[]_AC_LANG_ABBREV[]_sizeof_$1])
   AC_CACHE_CHECK([for size of $1], [CACHEVAR],dnl
    [AC_COMPUTE_INT([CACHEVAR], [sizeof($1)],dnl
      [m4_default($2, AC_INCLUDES_DEFAULT)], [AS_VAR_SET([CACHEVAR], [failed])])])
   AS_IF([test x"AS_VAR_GET([CACHEVAR])" != xfailed],dnl
    AC_DEFINE_UNQUOTED(AS_TR_CPP([LC_SIZEOF_$1]), [AS_VAR_GET([CACHEVAR])],dnl
      [size of $1])
    AC_DEFINE_UNQUOTED(AS_TR_CPP([HAVE_$1]), [1], [define if $1 is available]))
   AS_VAR_POPDEF([CACHEVAR])
])dnl LC_CHECK_TYPE_SIZE
